Solving a specific Sturm-Liouville problem

We will solve the following problem:

$-y''(x) = \lambda\, y(x)$

$y(0) - y'(0) = 0$

$y(\pi) = 0$

Negative eigenvalues

Set $\lambda = -\mu^2$, with $\mu > 0$

```
ExpToTrig[DSolve[y''[x] == μ^2 y[x], y[x], x]]
```

```
{{y[x] → C[1] Cosh[x μ] + C[2] Cosh[x μ] + C[1] Sinh[x μ] - C[2] Sinh[x μ]}}
```

Thus the general solution is

```
gsol = C[1] Cosh[x μ] + C[2] Sinh[x μ]
```

```
C[1] Cosh[x μ] + C[2] Sinh[x μ]
```

```
gsol /. {x → 0}
```

```
C[1]
```

```
D[gsol, x] /. {x → 0}
```

```
μ C[2]
```

Hence the first boundary condition is

```
(gsol /. {x → 0}) - (D[gsol, x] /. {x → 0})
```

```
C[1] - μ C[2]
```

The second boundary condition is

```
(gsol /. {x → Pi})
```

```
C[1] Cosh[π μ] + C[2] Sinh[π μ]
```

We need a nontrivial solution for C[1] and C[2] of the system

```
C[1] - μ C[2]  = 0
C[1] Cosh[π μ] + C[2] Sinh[π μ]  = 0
```

The determinant of this system is

```
Det[{{1, -μ}, {Cosh[π μ], Sinh[π μ]}}]
```

```
μ Cosh[π μ] + Sinh[π μ]
```

Since the nontrivial solution exixts only when the determinant is 0, we need to find μ-intercepts of the above function

```
Plot[μ Cosh[π μ] + Sinh[π μ], {μ, 0, 10}, PlotRange → {0, 10}]
```



This function does not have positive μ-intercepts. Thus our problem has no negative eigenvalues.

Zero eigenvalue

```
DSolve[y''[x] == 0, y[x], x]
```

```
{{y[x] → C[1] + x C[2]}}
```

The general solution is

```
gsol0 = C[1] + x C[2]
```

```
C[1] + x C[2]
```

```
gsol0 /. {x → 0}
```

```
C[1]
```

```
D[gsol0, x] /. {x → 0}
```

```
C[2]
```

The first boundary condition is

```
(gsol0 /. {x → 0}) - (D[gsol0, x] /. {x → 0})
```

```
C[1] - C[2]
```

The second boundary condition is

```
(gsol0 /. {x → Pi})
```

```
C[1] + π C[2]
```

Since the determinant

```
Det[{{1, -1}, {1, π}}]
```

```
1 + π
```

is nonzero the number zero is not an eigenvalue of our problem.

Positive eigenvalues

Set $\lambda = \mu^2$, with $\mu > 0$

```
DSolve[-y''[x] == μ^2 y[x], y[x], x]
```

```
{{y[x] → C[1] Cos[x μ] + C[2] Sin[x μ]}}
```

The general solution is

```
gsolp = C[1] Cos[x μ] + C[2] Sin[x μ]
```

```
C[1] Cos[x μ] + C[2] Sin[x μ]
```

The first boundary condition is

```
(gsolp /. {x → 0}) - (D[gsolp, x] /. {x → 0})
```

```
C[1] - μ C[2]
```

The second boundary condition is

```
(gsolp /. {x → Pi})
```

```
C[1] Cos[π μ] + C[2] Sin[π μ]
```

The determinant of the system for C[1] and C[2] is

```
Det[{{1, -μ}, {Cos[π μ], Sin[π μ]}}]
```

```
μ Cos[π μ] + Sin[π μ]
```

The system for C[1] and C[2] has nontrivial solution if and only if the above determinant is 0. Hence we need to find those μ-s for which the above determinant equals 0. Visualy:

```
Plot[μ Cos[π μ] + Sin[π μ], {μ, 0, 12},
 PlotRange → {-10, 10}, Ticks → {Range[0, 12], Automatic}]
```



Or, we need to find the intersections of the following functions

```
Plot[{-μ, Sin[π μ]/Cos[π μ]}, {μ, 0, 12}, PlotRange → {-10, 10},
 Ticks → {Range[0, 12], Automatic}, Exclusions → Range[1/2, 13, 1]]
```



To find approximate solutions for μ we use

```
FindRoot[μ Cos[π μ] + Sin[π μ] == 0, {μ, .7}]
```

```
{μ → 0.787637}
```

```
μ /. FindRoot[μ Cos[π μ] + Sin[π μ] == 0, {μ, .7}]
```

```
0.787637
```

Here is the first 13 values for μ placed in a table

```
mus13 = Table[μ /. FindRoot[μ Cos[π μ] + Sin[π μ] == 0, {μ, k + .7}], {k, 0, 12}]
```

```
{0.787637, 1.67161, 2.61621, 3.58655, 4.56859, 5.55668,
 6.54824, 7.54196, 8.53712, 9.53327, 10.5301, 11.5275, 12.5254}
```

Verify graphically:

```
Plot[μ Cos[π μ] + Sin[π μ], {μ, 0, 12}, PlotRange → {-10, 10},
 Ticks → {Range[0, 12], Automatic}, Epilog → {{Red, Point[{#, 0}] & /@ mus13}}]
```



```
Plot[{-μ, Sin[π μ]/Cos[π μ]}, {μ, 0, 12}, PlotRange → {-13, 4}, Ticks → {Range[0, 12], Automatic},
 Exclusions → Range[1/2, 13, 1], Epilog → {{Red, Point[{#, -#}] & /@ mus13}}]
```



These are corresponding 13 eigenfunction

```
yy[x_, k_] := 1 Cos[x mus13[[k]]] + 1/mus13[[k]] Sin[x mus13[[k]]]
Manipulate[Plot[yy[x, k], {x, 0, Pi}, PlotRange → {-2, 2}], {k, Range[13], Setter}]
```



Now we verify that the distinct eigenfunctions are mutually orthogonal:

```
MatrixForm[Chop[Table[Integrate[yy[x, j] yy[x, k], {x, 0, Pi}], {j, 1, 13}, {k, 1, 13}]]]
```

4.90878	0	0	0	0	0	0	0	0	0	0
0	2.31188	0	0	0	0	0	0	0	0	0
0	0	1.87334	0	0	0	0	0	0	0	0
0	0	0	1.73178	0	0	0	0	0	0	0
0	0	0	0	1.67001	0	0	0	0	0	0
0	0	0	0	0	1.63786	0	0	0	0	0
0	0	0	0	0	0	1.61909	0	0	0	0
0	0	0	0	0	0	0	1.6072	0	0	0
0	0	0	0	0	0	0	0	1.59921	0	0
0	0	0	0	0	0	0	0	0	1.59358	0
0	0	0	0	0	0	0	0	0	0	1.58947
0	0	0	0	0	0	0	0	0	0	0
0	0	0	0	0	0	0	0	0	0	0

Next we choose a function that we want to approximate with a linear combination of eigenfunctions. We choose a polynomial which satisfies the boundary conditions:

```
((a + b x) (x - Pi) /. {x → 0}) - (D[(a + b x) (x - Pi), x] /. {x → 0})
```

$-a - a\pi + b\pi$

```
Solve[-a - a π + b π == 0, b]
```

$\left\{\left\{b \to \frac{a + a\pi}{\pi}\right\}\right\}$

```
{{b → (Pi + Pi π)/π}}
```

$\left\{\left\{b \to \frac{\pi + \pi^2}{\pi}\right\}\right\}$

```
(Pi + Pi π)/π
```

$\frac{\pi + \pi^2}{\pi}$

```
(Pi + (1 + Pi) x) (x - Pi)
```

$(-\pi + x)\,(\pi + (1 + \pi)\,x)$

```
FullSimplify[((-π + x) (π + (1 + π) x) /. {x → 0}) - (D[(-π + x) (π + (1 + π) x), x] /. {x → 0})]
```

0

```
ff[x_] := (-π + x) (π + (1 + π) x)
```

The coefficients in the approximation of ff using a linear combination of eigenfunctions are

```
aa13 =
 Table[Integrate[ff[x] yy[x, j], {x, 0, Pi}] / Integrate[ yy[x, j]^2, {x, 0, Pi}], {j, 1, 13}]
```

{-9.96566, 0.434963, -0.358727, 0.0787226, -0.0646322, 0.0246452, -0.0212139,
 0.0105259, -0.00935652, 0.00540288, -0.00490709, 0.00312574, -0.0028821}

Here is the approximation

```
Appff[x_] = Sum[aa13[[j]] yy[x, j], {j, 1, 13}]
```

```
-9.96566 (Cos[0.787637 x] + 1.26962 Sin[0.787637 x]) +
 0.434963 (Cos[1.67161 x] + 0.598227 Sin[1.67161 x]) -
 0.358727 (Cos[2.61621 x] + 0.382232 Sin[2.61621 x]) +
 0.0787226 (Cos[3.58655 x] + 0.278819 Sin[3.58655 x]) -
 0.0646322 (Cos[4.56859 x] + 0.218886 Sin[4.56859 x]) +
 0.0246452 (Cos[5.55668 x] + 0.179964 Sin[5.55668 x]) -
 0.0212139 (Cos[6.54824 x] + 0.152713 Sin[6.54824 x]) +
 0.0105259 (Cos[7.54196 x] + 0.132592 Sin[7.54196 x]) -
 0.00935652 (Cos[8.53712 x] + 0.117136 Sin[8.53712 x]) +
 0.00540288 (Cos[9.53327 x] + 0.104896 Sin[9.53327 x]) -
 0.00490709 (Cos[10.5301 x] + 0.0949655 Sin[10.5301 x]) +
 0.00312574 (Cos[11.5275 x] + 0.0867487 Sin[11.5275 x]) -
 0.0028821 (Cos[12.5254 x] + 0.079838 Sin[12.5254 x])
```

How good is our approximation?

```
Plot[{ff[x], Appff[x]}, {x, 0, Pi},
 PlotStyle → {{Blue, Thickness[0.01]}, {Red, Thickness[0.005]}}]
```



Now assume that the function ff is the initial shape of a string with appropriate coefficients. Then the oscillations of this string are well approximated with the function

```
uu[x_, t_] = Sum[aa13[[j]] yy[x, j] Cos[mus13[[j]] t], {j, 1, 13}]
```

```
-9.96566 Cos[0.787637 t] (Cos[0.787637 x] + 1.26962 Sin[0.787637 x]) +
 0.434963 Cos[1.67161 t] (Cos[1.67161 x] + 0.598227 Sin[1.67161 x]) -
 0.358727 Cos[2.61621 t] (Cos[2.61621 x] + 0.382232 Sin[2.61621 x]) +
 0.0787226 Cos[3.58655 t] (Cos[3.58655 x] + 0.278819 Sin[3.58655 x]) -
 0.0646322 Cos[4.56859 t] (Cos[4.56859 x] + 0.218886 Sin[4.56859 x]) +
 0.0246452 Cos[5.55668 t] (Cos[5.55668 x] + 0.179964 Sin[5.55668 x]) -
 0.0212139 Cos[6.54824 t] (Cos[6.54824 x] + 0.152713 Sin[6.54824 x]) +
 0.0105259 Cos[7.54196 t] (Cos[7.54196 x] + 0.132592 Sin[7.54196 x]) -
 0.00935652 Cos[8.53712 t] (Cos[8.53712 x] + 0.117136 Sin[8.53712 x]) +
 0.00540288 Cos[9.53327 t] (Cos[9.53327 x] + 0.104896 Sin[9.53327 x]) -
 0.00490709 Cos[10.5301 t] (Cos[10.5301 x] + 0.0949655 Sin[10.5301 x]) +
 0.00312574 Cos[11.5275 t] (Cos[11.5275 x] + 0.0867487 Sin[11.5275 x]) -
 0.0028821 Cos[12.5254 t] (Cos[12.5254 x] + 0.079838 Sin[12.5254 x])
```

```
Manipulate[Plot[uu[x, t], {x, 0, Pi}, PlotRange → {{-1.1, Pi + .1}, {-20, 20}},
  PlotStyle → {{Blue, Thickness[0.01]}}, Axes → False, Frame → True,
  FrameTicks → {{-1, 0, Pi / 2, Pi}, {}, {}, {}}, AspectRatio → 1 / 5, Epilog →
   {{Red, Thickness[0.01], Line[{{-1, 0}, {0, uu[0, t]}}]}}, ImageSize → 400], {t, 0, 8.18}]
```



Let us approximate one more function, again polynomial but with two roots

```
((a + b x) (x - 2) (x - Pi) /. {x → 0}) - (D[(a + b x) (x - 2) (x - Pi), x] /. {x → 0})
```

$2 a + 3 a \pi - 2 b \pi$

```
Solve[2 a + 3 a π - 2 b π == 0, b]
```

$\left\{\left\{b \to \frac{2 a + 3 a \pi}{2 \pi}\right\}\right\}$

$\left\{\left\{b \to \frac{2 + 3 \pi}{2}\right\}\right\}$

$\left\{\left\{b \to \frac{1}{2} (2 + 3 \pi)\right\}\right\}$

$\left(Pi + \frac{2 + 3 \pi}{2} x\right) (x - 2) (x - Pi)$

$(-2 + x) (-\pi + x) \left(\pi + \frac{1}{2} (2 + 3 \pi) x\right)$

```
FullSimplify[
 (((Pi + (2 + 3 π)/2 x) (x - 2) (x - Pi)) /. {x → 0}) - (D[(Pi + (2 + 3 π)/2 x) (x - 2) (x - Pi), x] /. {x → 0})]
```

0

```
gg[x_] := (Pi + (2 + 3 π)/2 x) (x - 2) (x - Pi)
```

The coefficients in the approximation of ff using a linear combination of eigenfunctions are

```
aa13g =
 Table[Integrate[gg[x] yy[x, j], {x, 0, Pi}] / Integrate[ yy[x, j]^2, {x, 0, Pi}], {j, 1, 13}]
```

```
{8.70422, 10.7634, -0.773925, 1.02016, -0.235739, 0.255198, -0.0937333,
 0.0974657, -0.0456623, 0.0468012, -0.0254482, 0.0259042, -0.0155717}
```

Here is the approximation

```
Appgg[x_] = Sum[aa13g[[j]] yy[x, j], {j, 1, 13}]
```

```
8.70422 (Cos[0.787637 x] + 1.26962 Sin[0.787637 x]) +
 10.7634 (Cos[1.67161 x] + 0.598227 Sin[1.67161 x]) -
 0.773925 (Cos[2.61621 x] + 0.382232 Sin[2.61621 x]) +
 1.02016 (Cos[3.58655 x] + 0.278819 Sin[3.58655 x]) -
 0.235739 (Cos[4.56859 x] + 0.218886 Sin[4.56859 x]) +
 0.255198 (Cos[5.55668 x] + 0.179964 Sin[5.55668 x]) -
 0.0937333 (Cos[6.54824 x] + 0.152713 Sin[6.54824 x]) +
 0.0974657 (Cos[7.54196 x] + 0.132592 Sin[7.54196 x]) -
 0.0456623 (Cos[8.53712 x] + 0.117136 Sin[8.53712 x]) +
 0.0468012 (Cos[9.53327 x] + 0.104896 Sin[9.53327 x]) -
 0.0254482 (Cos[10.5301 x] + 0.0949655 Sin[10.5301 x]) +
 0.0259042 (Cos[11.5275 x] + 0.0867487 Sin[11.5275 x]) -
 0.0155717 (Cos[12.5254 x] + 0.079838 Sin[12.5254 x])
```

How good is our approximation?

```
Plot[{gg[x], Appgg[x]}, {x, 0, Pi},
 PlotStyle → {{Blue, Thickness[0.01]}, {Red, Thickness[0.005]}}]
```



Now assume that the function gg is the initial shape of a string with appropriate coefficients. Then the oscillations of this string are well approximated with the function

```
uug[x_, t_] = Sum[aa13g[[j]] yy[x, j] Cos[mus13[[j]] t], {j, 1, 13}]
```

```
8.70422 Cos[0.787637 t] (Cos[0.787637 x] + 1.26962 Sin[0.787637 x]) +
 10.7634 Cos[1.67161 t] (Cos[1.67161 x] + 0.598227 Sin[1.67161 x]) -
 0.773925 Cos[2.61621 t] (Cos[2.61621 x] + 0.382232 Sin[2.61621 x]) +
 1.02016 Cos[3.58655 t] (Cos[3.58655 x] + 0.278819 Sin[3.58655 x]) -
 0.235739 Cos[4.56859 t] (Cos[4.56859 x] + 0.218886 Sin[4.56859 x]) +
 0.255198 Cos[5.55668 t] (Cos[5.55668 x] + 0.179964 Sin[5.55668 x]) -
 0.0937333 Cos[6.54824 t] (Cos[6.54824 x] + 0.152713 Sin[6.54824 x]) +
 0.0974657 Cos[7.54196 t] (Cos[7.54196 x] + 0.132592 Sin[7.54196 x]) -
 0.0456623 Cos[8.53712 t] (Cos[8.53712 x] + 0.117136 Sin[8.53712 x]) +
 0.0468012 Cos[9.53327 t] (Cos[9.53327 x] + 0.104896 Sin[9.53327 x]) -
 0.0254482 Cos[10.5301 t] (Cos[10.5301 x] + 0.0949655 Sin[10.5301 x]) +
 0.0259042 Cos[11.5275 t] (Cos[11.5275 x] + 0.0867487 Sin[11.5275 x]) -
 0.0155717 Cos[12.5254 t] (Cos[12.5254 x] + 0.079838 Sin[12.5254 x])
```

```
Manipulate[Plot[uug[x, t], {x, 0, Pi}, PlotRange → {{-1.1, Pi + .1}, {-27, 27}},
  PlotStyle → {{Blue, Thickness[0.01]}}, Axes → False, Frame → True,
  FrameTicks → {{-1, 0, Pi / 2, Pi}, {}, {}, {}}, AspectRatio → 1 / 5, Epilog →
   {{Red, Thickness[0.01], Line[{{-1, 0}, {0, uug[0, t]}}]}}, ImageSize → 400], {t, 0, 15.1}]
```